Example 8.3

Example (8.1.3) is solved in Maple below:

```
> restart:with(inttrans):with(plots):
> eq:=diff(u(x,t),t)=diff(u(x,t),x$2);
```

$$eq := \frac{\partial}{\partial t}\, u(x,t) = \frac{\partial^2}{\partial x^2}\, u(x,t) \tag{1}$$

```
> u(x,0):=sin(Pi*x);
```

$$u(x,0) := \sin(\pi\, x) \tag{2}$$

```
> bc1:=u(x,t)=0;
```

$$bc1 := u(x,t) = 0 \tag{3}$$

```
> bc2:=u(x,t)=0;
```

$$bc2 := u(x,t) = 0 \tag{4}$$

The governing equation and the boundary conditions are converted to the Laplace domain:

```
> eqs:=laplace(eq,t,s):
> eqs:=subs(laplace(u(x,t),t,s)=U(x),eqs);
```

$$eqs := s\, U(x) - \sin(\pi\, x) = \frac{d^2}{dx^2}\, U(x) \tag{5}$$

```
> bc1:=laplace(bc1,t,s):
> bc1:=subs(laplace(u(x,t),t,s)=U(0),bc1);
```

$$bc1 := U(0) = 0 \tag{6}$$

```
> bc2:=laplace(bc2,t,s):
> bc2:=subs(laplace(u(x,t),t,s)=U(1),bc2);
```

$$bc2 := U(1) = 0 \tag{7}$$

The governing equation in the Laplace domain is solved as:

```
> dsolve(eqs,U(x));
```

$$U(x) = e^{\sqrt{s}\, x}\, \_C2 + e^{-\sqrt{s}\, x}\, \_C1 + \frac{\sin(\pi\, x)}{s + \pi^2} \tag{8}$$

The governing equation is solved with the boundary conditions as:

```
> U(x):=rhs(dsolve({eqs,bc1,bc2},U(x)));
```

$$U(x) := \frac{\sin(\pi\, x)}{s + \pi^2} \tag{9}$$

The solution obtained is converted to the time domain as:

```
> u:=invlaplace(U(x),s,t);
```

$$u := \sin(\pi\, x)\, e^{-\pi^2 t} \tag{10}$$

```
> plot3d(u,x=1..0,t=0..0.2,axes=boxed,title="Figure Exp. 8.5.",
  labels=[x,t,"u"],orientation=[120,60]);
```




```
> plot([subs(t=0,u),subs(t=0.05,u),subs(t=0.1,u),subs(t=0.2,u)],x=
  0..1,title="Figure 8.6.",axes=boxed,thickness=5,labels=[x,"u"]);
```




```
>
```

In all the examples previously discussed, until now the boundary conditions did not involve derivatives. When there is a derivative in the boundary condition it has to be taken care of while applying the

Laplace transform as shown in the next example.